
The Forest
ZX Spectrum version, August 1983

This document is a collection of copies of the original design notes, source code
print-outs, diagrams and publication informaion for the second published version of
The Forest. It represents an archive of what was possible in a machine using an 8-bit
microprocessor (Z80) and only 48 kilobytes of programmable memory.

My first personal computer was a TRS-80 Model 1 Level II and I have written
separately (a document similar to this) about how the first version of The Forest was
written for that and published in March 1983.

I then bought a Sinclair ZX Spectrum. It also used the Z80 processor, so much of my
code was immediately transferable. However the Spectrum had a completely
different way of displaying graphics, and in colour rather than monochrome. It also
had a rather different flavour of BASIC for the higher level programming.

The machine had an interpreter for the programming language BASIC. That was
stored in read-only memory (ROM) which was immediately available when power
was switched on. Programs could either be typed into the remaining random-access
memory (RAM) or be loaded from audio cassette tapes. One such program was an
editor/assembler which made it possible to write additional programs in the Z80
machine code and save those on audio tape for reloading.

This version of The Forest was published by Phipps Associates, a company in South
London.

The Forest was one of the earliest attempts to make a simulation of a sport:
orienteering. That is, cross-country navigation on foot with map and compass to visit
set control points in a planned order.

My particular innovation was to devise a way to generate limitless terrain from a very
small amount of program code. The terrain was contoured and had several different
types of vegetation plus towns and lakes. It had various kinds of point features dotted
around, such as boulders and ponds.

Although I demonstrated that detailed terrain could be generated, the display
hardware in those days was extremely limited. So a version of the program was
made to output all of the map data as if surveying a real forest. Then the map could
be drawn and printed just as we did for real orienteering events (I had been doing
some of that since 1973). The published program was therefore accompanied by a
full-colour printed map. This was seen as an advantage because it made it difficult to
copy the product. People did not have colour printer/copiers in those days either.

The Z80 is fundamentally an 8-bit processor but with 16-bit memory addressing,
allowing up to 64 kilobytes of memory. My Spectrum had the full 64kb, comprising
16kb of ROM plus 48kb of RAM. Some of the 8-bit registers could be programmed as
pairs to facilitate 16-bit integer arithmetic.

For The Forest I had no access to any floating-point arithmetic so I had to make
some interesting design choices in the first, TRS-80, version which I kept for the
Spectrum version. For the (x, y) position of the orienteer on the map I used 3-byte

fixed-point numbers: 2 bytes for the integer part and 1 byte after the decimal point. In
my design notes it is written as HL.A to use the 16-bit HL register pair and the 8-bit
accumulator. This enabled the map to have a width and height of 65km before
repeating but the orienteer could be positioned to 1/256 of a metre. That was
necessary to allow forward motion on any bearing, using sines and cosines. Which
raised another point: how to do trigonometry without floating point? The answer
involved 2 more design choices. Firstly compass bearings were “b-degrees” of which
I had 256 to a complete circle: one byte’s worth. Bearings therefore had a resolution
of 360 / 256 or about 1.4 degrees, more than accurate enough for an orienteer
reading a compass. Then sines and cosines were looked up in a table of byte values,
each being 128 times the true value and so represented as signed 1-byte integers.
The table only needed 64 entries because values in the 4 quadrants are simply
related. Also the table as read from the beginning gave sines but read backwards
from the end it gave cosines. All to save memory and it worked very well.

The BASIC ROM had a floating-point calculator using 5-byte numbers (1 byte for the
exponent). When I got the Melbourne House book of its full disassembly it was made
clear that the calculator was not designed for speed. It would not have been fast
enough for my purposes. Remember also that the clock speed was only about 4MHz.

The maps

The next 2 pages show the printed orienteering maps which accompanied the
published program. It has to be admitted that in the first one the contours are not very
realistic. That was only the map issued for the TRS-80 version. While converting the
program to run on the Spectrum I developed a much more realistic terrain generator
which resulted in the second map, “The Complex Forest”. As will be shown in the
listings this involved substituting a chunk of code in the terrain generator and
providing a switching mechanism in the BASIC program. There are pages below
which explain the difference in detail. (The same could be done retrospectively in the
TRS-80 version if anyone is interested. I have discovered that there are enthusiasts
still running those machines and resurrecting code such as mine.)

Graham Relf August 2022

Many of the pages in this document are snapshots of my original handwritten notes.

But later you will see that I have provided most assembler routines as text, to
facilitate possible reuse. If this typed-in code doesn’t work I think the most likely
cause will be that I have got addresses wrong in the various EQU lines. I have
included some pages with memory maps which could be checked against the EQUs.

; New GRAFPRINT
ORG 65200 ; Position just below UDG. (User-Defined Graphics)

ADATT: DEFW 0 ; 65200 Pointer to display position in attr file
; (22528..23295)

ADGRA: DEFW 0 ; 65202 Pointer to first byte of graphics string
GSHOW: LD DE,(ADGRA) ; 65204 Point to first byte of graphic

LD HL,(ADATT) ; 65211 Point to display position in attr file
G0: PUSH HL ; Save attr file address

LD A,H ; Look at MSB:
CP 88 ; If < 88
JR C,OFFSC ; then off top of screen
CP 89 ; else if < 89
JR C,BLK1 ; then in top block of screen
CP 90 ; else if < 90
JR C,BLK2 ; then in middle block of screen
CP 91 ; else if >= 91
JR NC,OFFSC ; then off bottom of screen

BLK3: LD BC,2560 ; Offset from bottom attr block to display file
JR COMMN ; Join common path

BLK2: LD BC,4352 ; Offset from middle attr block to display file
JR COMMN ; Join common path

BLK1: LD BC,6144 ; Offset from top attr block to display file
COMMN: LD A,(DE) ; Get 1st byte of character, the attribute code

LD (HL),A ; Set attributes for this character
AND A ; Clear carry
SBC HL,BC ; Point to corresponding position in display file
INC DE ; Point to first pixel byte this character
LD B,8 ; Count 8 pixel bytes

GLOOP: LD A,(DE) ; Get pixel byte and
LD (HL),A ; display it
DEC BC ; If 8 bytes done for this character
JR Z,EXITL ; then exit loop
INC H ; Point to next display byte this char, 256 forward
INC DE ; Point to next pixel byte in character
JR GLOOP ; Next pixel byte

OFFSC: LD B,8 ; Off screen: loop to skip to 8th pixel this char
OLOOP: INC DE

DJNZ OLOOP
EXITL: POP HL ; Restore pointer to attr file

INC DE ; Point to last byte this char, the offset to next
in attr file

LD A,(DE) ; Get offset
CP 0 ; If zero
RET Z ; then end of string
LD B,0 ; Clear B
CP 128 ; If offset < 0
JR C,POSOK
DEC B ; then B = 255, ie extend sign for BC

POSOK: LD C,A ; BC = (signed) offset
ADD HL,BC ; Add offset to attr file ptr to get new display posn
INC DE ; Point to 1st byte of next character
JR G0 ; Do next character

ORG 63850 ; CLEAR SCREEN
SDF: EQU 16384 ; Start of display file
SAF: EQU 22528 ; Start of attribute file
ATTR0: DEFB 0 ; Attributes byte for cleared screen: poke a value
ZERO: DEFB 0 ; Value for display file
CLS: LD BC,32*24 ; 24 lines to set in attr file

LD DE,SAF ; Destination for move
LD HL,ATTR0 ; Source for move
CALL CLOOP ; Move
LD BC,32*192 ; 192 lines to clear in display file
LD DE,SDF ; Destination for move
LD HL,ZERO ; Source for move

CLOOP: LDI ; Move
DEC HL ; Keep pointing to single source byte
JP PE,CLOOP ; Repeat until BC = 0
RET

ORG 63745 ; TEXTPRINT
ATTR: DEFB 0 ; Attributes to be used for string
ADATT: DEFW 0 ; Pointer to display position in attr file
ADTEX: DEFW 0 ; Pointer to 1st byte of text string
TSHOW: LD DE,(ADTEX) ; Point to string

LD HL,(ADATT) ; Point to attr file
T0: PUSH HL ; Save attr file pointer

LD A,H ; Look at MSB
CP 88 ; If < 88
JR C,OFFSC ; then off top of screen
CP 89 ; else if < 89
JR C,BLK1 ; then in top block
CP 90 ; else if < 90
JR C,BLK2 ; then in middle block
CP 91 ; else if >= 91
JR NC,OFFSC ; then off bottom of screen
LD BC,2560 ; Offset to display file, bottom block
JR COMMN

BLK2: LD BC,4352 ; Offset to display file, middle block
JR COMMN

BLK1: LD BC,6144 ; Offset to display file, top block
COMMN: LD A,(DE) ; Get ASCII byte from string

CP 0 ; If null
JR Z,ENDT ; then end of string
LD A,(ATTR) ; Get attribute byte for string
LD (HL),A ; Set attributes for this character
AND A ; Clear carry
SBC HL,BC ; Offset pointer to display file
PUSH DE ; Save string pointer
EX DE,HL ; DE = display file pointer
LD L,(HL) ; Get ASCII code byte
LD H,0 ; HL = ASCII code byte
ADD HL,HL
ADD HL,HL

ADD HL,HL ; HL = 8 * ASCII code byte
LD BC,15360 ; Base addr of std char set in ROM (=15616 - 32 * 8)
ADD HL,BC ; HL points to 1st pixel byte in ROM for ASCII char
LD B,8 ; Count 8 pixel bytes

TLOOP: LD A,(HL) ; Get pixel byte
LD (DE),A ; Display it
DEC B ; If 8 bytes displayed
JR Z,EXITL ; then character done
INC D ; Point to next display file position this char, 256

forward
INC HL ; Point to next pixel byte
JR TLOOP ; Next byte

EXITL: POP DE ; Restore string pointer
OFFSC: INC DE ; Next char in string

POP HL ; Restore attr file pointer
INC HL ; Next attr file display position
JR T0 ; Next character

ENDT: POP HL ; Tidy the stack
RET

Format for stored text string: 1 byte ASCII code for each character
followed by zero byte to show end of string.

ORG 63200 ; INIT (Title page & initialise global data)
T0: EQU 63757
G0: EQU 65210
CLS: EQU 63852
ATTR0: EQU 63850
TATTR: EQU 63745
TITLE: DEFM "The Forest"

DEFB 0
COPYR: DEFM "Copyright © Graham T Relf 1983 " ; 32 characters
DISTR: DEFS 32,32 ; Spare line for distributor name
PKEY: DEFM "Press any key to start"

DEFB 0
INIT: LD A,12 ; Paper blue, ink green

LD (ATTR0),A
CALL CLS
LD A,15 ; Paper blue, ink white
LD (TATTR),A
LD DE,TITLE
LD HL,23168 ; Line 20, column 0
CALL T0
LD DE,COPYR
LD HL,23200 ; Line 21, column 0
CALL T0 ; 3 lines just run on
LD DE,64600 ; GRUN
LD HL,23138 ; Line 18, column 2
CALL G0 ; Display tall tree
LD DE,64400 ; GFIR
LD HL,23141 ; Line 19, column 5
CALL G0 ; Display short tree



The Forest - ZX Spectrum page 22

LD DE,64400 ; GFIR
LD HL,23144 ; Line 19, column 8
CALL G0
LD DE,64100 ; GMOOR
LD HL,23147 ; Line 19, column11
CALL G0 ; Display moor symbol
LD DE,63900 ; GLAKE
LD HL,23182 ; Line 20, column 14
CALL G0
LD DE,63900 ; GLAKE
LD HL,23185 ; Line 20, column 17
CALL G0
LD DE,64200 ; GTOWN
LD HL,23156 ; Line 19, column20
CALL G0
LD DE,64400 ; GFIR
LD HL,23159 ; Line 19, column 23
CALL G0
LD DE,64000 ; GFLAG
LD HL,23162 ; Line 19, column 26
CALL G0
LD DE,64600 ; GRUN
LD HL,23165 ; Line 19, column 29
CALL G0
LD HL,CONST
LD DE,VARS
LD BC,NCONS
LDIR ; Set first NCONS variables to values in CONST
RET

VARS: DEFS 64
CONST: DEFB 64 ; Initial values: bearing 90 degrees = 64b

DEFB 0 ; sin b = 0
DEFB 128 ; cos b = 1 (times 128)
DEFB 0 ; Observer's start x = 1366.0
DEFW 1366
DEFB 0 ; Observer's start y = 2332.0
DEFW 2332
DEFS 12 ; 12 bytes do not need initialising
DEFW 1438 ; x of finish
DEFW 2272 ; y of finish
DEFB 0 ; Clear the finished flag

NCONS: EQU 26 ; 26 bytes to move for initialisation
END

See later for details of how the terrain generator is modified for the improved
“Complex Forest”.

ORG 62580 ;TERRA - the original terrain generator
FXH: EQU 63440
LXH: EQU FXH + 1
HXH: EQU FXH + 2
FYH: EQU FXH + 3
LYH: EQU FXH + 4
HYH: EQU FXH + 5
FHH: EQU FXH + 6
LHH: EQU FXH + 7
HHH: EQU FXH + 8
TG: EQU FXH + 18
TERSY: EQU FXH + 19
TERRA: CALL HEIGH

LD A,(FH)
LD (FHH),A
LD HL,(LH)
LD (LHH),HL
LD A,H
CP 0
JR NZ,LAND
LD A,L
CP 204 ;Lake height + bank
JR C,TLAKE ;Lake here

LAND: XOR A ;A = 0
LD (FXH),A ;Attributes to constant
LD (FYH),A ;over 1m cells
CALL ATTRI ;NB: HEIGH left IX pointing to TABA1
LD A,(HH)
CP 0
JR NZ,CNTRY
LD A,(LH)
CP 150
JR C, TTOWN ;Town here

CNTRY: CALL ATTRI ;NB: Last ATTRI left IX pointing to TABA2
LD A,(HH)
CP 0
JR NZ,WOOD
LD A,(LH)
CP 255
JR C,OPEN

WOOD: CALL ATTRI ;NB: Last ATTRI left IX pointing to TABA3
LD A,(HH)
CP 0
JR NZ,TRUN ;Runnable wood here
LD A,(LH)
CP 255
JR C,THICK ;Thicket here

TRUN: LD A,0
LD (TERSY),A
JP FEATU

THICK: LD A,2
LD (TERSY),A
JP FEATU



The Forest - ZX Spectrum page 33

OPEN: CALL ATTRI ;NB: Last ATTRI left IX pointing to TABA3
LD A,(HH)
CP 0
JR NZ,TMOOR ;Moor here
LD A,(LH)
CP 255
JR C,TGRAS

TMOOR: LD A,4
LD (TERSY),A
JP FEATU

TGRAS: LD A,6
LD (TERSY),A
JP FEATU

TTOWN: LD A,8
LD (TERSY),A
JP FEATU

TLAKE: LD A,200 ;Set lake level so we cannot
LD (LHH),A
XOR A
LD (FHH),A
LD A,10
LD (TERSY),A
JP FEATU

;Height & attributes function:
HEIGH: LD IX,ABH ;Point to table of height coefficients
ATTRI: XOR A ;A = 0

LD (FH),A ;Initialise
LD (LH),A ;the
LD (HH),A ;sum.
LD B,5 ;No of coeff pairs in table

HLOO1: PUSH BC ;Save loop counter i
LD DE,(LXH) ;Get
LD A,(FXH) ;x-coordinate
LD B,A
CALL MULCO ;HL.A = a[i]*x, point IX to b[i]
PUSH HL
PUSH AF
LD DE,(LYH) ;Get
LD A,(FYH) ;y-coordinate
LD B,A
CALL MULCO ;HL.A = b[i] * y, point IX to a[i+1]
POP BC
POP DE ;DE.B = a[i] * x
ADD A,B
ADC HL,DE ;HL.A = a[i]x + b[i]y

NORM1: LD D,L
LD E,A ;D.E = result mod 256
LD A,D ;MSB
CP 127 ;If in right half of triangle
JR C,REPOS
LD A,E ;then fold over (ie, negate D.E)
CPL
ADD A,1

LD E,A
LD A,D
CPL
ADC A,0
LD D,A ;D.E = 256 - D.E

REPOS: LD HL,(FH) ;Add into overall sum in FH,LH,HH
ADD HL,DE
LD (FH),HL
LD A,(HH)
ADC A,0
LD (HH),A
POP BC ;Retrieve loop counter, i
DJNZ HLOO1 ;Next i
RET

;Multiply coord DE.B by signed coeff pointed by IX
;Result in HL.A
MULCO: XOR A ;A = 0

LD H,A
LD L, A
LD C,(IX+0) ;Get coeff a[i] or b[i]
BIT 7,C ;Sign?
JR NZ,SUBCO
CALL ADDBT
CALL ADDBT
CALL ADDBT
CALL ADDBT
CALL ADDBT
CALL ADDBT
CALL ADDBT ;7 times

ENDCO: RR H
RR L
RRA
INC IX ;Point to next coeff, b[i] or a[i+1]
RET

SUBCO: CALL SUBBT
CALL SUBBT
CALL SUBBT
CALL SUBBT
CALL SUBBT
CALL SUBBT
CALL SUBBT ;7 times
JR ENDCO

ADDBT: RR H
RR L
RRA
RR C
RET NC
ADD A,B
ADC HL,DE
RET

SUBBT: RR H
RR L
RRA

RR C
RET NC
SUB B
SBC HL,DE
RET

;Point features, flags & codes
FEATY: DEFB 0 ;Index to 1..63 array of probabilities of
;each feature type (0..11)
CODE: DEFB 32 ;2-letter flag code,

DEFB 32 ;spaces if no flag
DEFB 0

FEATU: LD A,(TERSY) ;Get observer's terrain type
CP 7
JR NC,NOFEA ;No features in lake or town
LD IX,TABA1-5 ;Re-use existing coeff tables
XOR A ;A = 0
LD (FXH),A ;Features to be visible in 8x8m area:
LD (FYH),A
LD A,(LXH)
AND 0F8H
LD (LXH),A
LD A,(LYH)
AND 0F8H
LD (LYH),A
CALL ATTRI ;Results in FH,LH,HH
LD A,(LH) ;NB: FH = 0 so can't use that
AND 097H ;10010111
CP 128
JR NZ,NOFEA
CALL ATTRI ;NB: Last ATTRI left IX at TABA2-5
LD A,(LH)
AND 63 ;Indexing 1..63 array
LD (FEATY),A
LD A,(LH)
AND 64
JR Z,NOCOD ;1 in 2 chance of a flag
CALL ATTRI ;NB: Last ATTRI left IX at TABA3-5
LD A,(HH)
AND 15 ;Range A..O [in practice A..C]
ADD A,65 ;To ASCII letter
LD (CODE),A
LD A,(LH)
AND 15 ;Range A..O
ADD A,65
LD (CODE+1),A
RET

NOFEA: XOR A ;A = 0
LD (FEATY),A

NOCOD: LD A,32
LD (CODE),A
LD (CODE+1),A
RET

;Local data:

FH: DEFB 0
LH: DEFB 0
HH: DEFB 0
;Tables of coefficients:
ABH: DEFB 200 ;a1

DEFB 3 ;b1
DEFB 165 ;a2
DEFB 37 ;b2
DEFB 127 ;a3
DEFB 96 ;b3
DEFB 101 ;a4
DEFB 202 ;b4
DEFB 13 ;a5
DEFB 157 ;b5

TABA1: DEFB 171
DEFB 131
DEFB 171
DEFB 140
DEFB 184
DEFB 106
DEFB 97
DEFB 126
DEFB 124
DEFB 96

TABA2: DEFB 152
DEFB 107
DEFB 153
DEFB 182
DEFB 68
DEFB 162
DEFB 118
DEFB 179
DEFB 158
DEFB 171

TABA3: DEFB 179
DEFB 172
DEFB 190
DEFB 108
DEFB 186
DEFB 101
DEFB 192
DEFB 192
DEFB 95

TABAD: DEFB 172
END

; ADS - address tables + 2 subroutines
ORG 62350

; Graphic for grass:
GRASS: DEFB 32 ; Paper green, ink black

DEFS 8 ; 8 zero bytes - pure paper
DEFB 1 ; Offset in attr file to next byte
DEFB 32
DEFS 8
DEFB 1
DEFB 32
DEFS 8
DEFB 0

; The next 2 tables are indexed by 2t, where t = terrain type, 0..5
ADSYM: DEFW 64600 ; Table of graphics

DEFW 64400
DEFW 64100
DEFW GRASS
DEFW 64200
DEFW 63900

ADOBS: DEFW 63600 ; Table of terrain messages
DEFW 63604
DEFW 63610
DEFW 63615
DEFW 63626
DEFW 63621

BEAR: EQU 63431
SINB: EQU 63432
SINCO: EQU 63125
TURND: LD A,(BEAR) ; Compute & store sin & cos of new bearing

CALL SINCO ; after a turn
LD HL,SINB
LD (HL),D



The Forest - ZX Spectrum page 38

INC HL
LD (HL),E
RET

; Multiply +ve 2-byte fixed-point no in D.E by 1-byte signed fraction in C
; [+/- .7 bits], to get signed 2-byte fixed-point in H.L [+/- 7Bits.byte].
; NB: D.E not changed
MUL212: LD A,1 ; Sign +ve until otherwise

BIT 7,C ; If sign(C) < 0
JR Z,POS212
LD A,C ; then
NEG ; complement and
LD C,A ; set -ve sign
LD A,255 ; for result

POS212: LD B,7 ; 7 bits left in C to shift & add
LD HL,0
AND A ; Clear carry

LO212: RR H
RR L
OR C
JR NC,EN212 ; If bit set at right of C
ADD HL,DE ; then add DE into result. NB: Carry affected

EN212: DJNZ LO212 ; NB: Loop test preserves carry
RR H ; Shift to put any
RR L ; carry into result
CP 255 ; If sign (in A) +ve
RET NZ ; then done

NEG2: LD A,L ; else take
CPL ; twos complement
ADD A,1 ; of result
LD L,A
LD A,H
CPL
CPL
ADC A,0
LD H,A
AND 80H ; Select bit 7 (sign)
RLCA
NEG
LD B,A ; HL >= 0 => B = 0 else B = -1
RET
END

; NEG2 may be used as general subroutine
; to complement HL

; SHOFEA - show point feature & flag
ORG 62495

T0: EQU 63757
SHOFE: LD HL,FIN1

LD B,12 ; 12 feature types, counter
SHLOO: CP (HL) ; If feature index in A > table entry

JR C,INC2 ; then this is the feature type
INC HL
INC HL ; Word-based table => 2 increments
DJNZ SHLOO
RET

INC2: LD DE,FIN2-FIN1
ADD HL,DE ; HL points to address of text string
LD E,(HL)
INC HL
LD D,(HL)
LD HL,23283 ; Point to attr file, line 23, column 19:
CALL T0 ; 22528 + 32 * 23 + 19
RET

; Index1: probability table for point features (see part 4)
FIN1: DEFW 11

DEFW 21
DEFW 24
DEFW 29
DEFW 31
DEFW 35
DEFW 42
DEFW 53
DEFW 55
DEFW 60
DEFW 62
DEFW 64

; Index 2: string addresses
FIN2: DEFW 63631 ; "Depression"

DEFW 63642 ; "Knoll"
DEFW 63648 ; "Niche"
DEFW 63722 ; "Pond"
DEFW 63727 ; "Water tank"
DEFW 63662 ; "Building"
DEFW 63709 ; "Rock outcrop"
DEFW 63654 ; "Boulder"
DEFW 63671 ; "Ruin"
DEFW 63698 ; "Mine shaft"
DEFW 63676 ; "Root stock"
DEFW 63687 ; "Sheep fold"
END

ORG 62050 ; DISPL - display current scene
CHLIN: DEFB 0
SCLIN: DEFB 0
TURN: CALL TURND
DISPL: LD A,12 ; Paper blue, ink green

LD (ATTR0),A
CALL CLS ; Clear screen
LD A,2 ; 1st col to display (tree can overhang col 1)
LD (COLUM),A
LD A,(BEAR) ; Get straight-ahead bearing
SUB 36 ; Look 4.5 symbol spaces left
LD (DBEAR),A

DLOO1: LD A,(DBEAR)
CALL SINCO ; DE = sin & cos of horizon direction
LD (SIND),DE ; Save both sin & cos
LD L,D ; Move sin to L
CALL DTRIG ; Compute rc * sin bt
LD DE,(FXT) ; Get observer's x in AD.E
LD A,(HXT)
ADD HL,DE
ADC A,C ; xh = xt + rc * sin bt
LD (HXH),A ; Horizon position
LD (FXH),HL
LD HL,(SIND) ; Get sin & cos of horizon direction
CALL DTRIG ; Compute rc * cos bt
LD DE,(FYT) ; Get observer's y in AD.E
LD A,(HYT)
ADD HL,DE
ADC A,C
LD (HYH),A
LD (FYH),HL
CALL TERRA ; Get height, terrain, feature for horizon point
LD A,(FHH)
LD C,A
LD A,(FHG)
SUB C
LD HL,(LHG)
LD DE,(LHH)
SBC HL,DE ; HL.A = (hg-hh)*sh = ht difference, sh = 1 => range

-23..24
LD BC,24
ADD HL,BC ; HL.A += 24, range 0..47
LD H,L
LD L,A ; H.L = ht diff, range 47..0
LD A,H
SRL A ; A = ht diff, range 23..0
LD (CHLIN),A ; Character-sized position
XOR A ; A = 0
LD (HFLAG),A ; No flag until shown otherwise
LD A,(CODE)
CP 32 ; Is flag visible from horizon point?
JR Z, NFLAG
LD A,(OBSCO)

CP 32 ; Has observer got a flag too?
JR NZ,NFLAG
LD A,1
LD (HFLAG),A ; Horizon flag to be shown

NFLAG: LD A,(TERSY) ; Get terrain type (x 2)
LD C,A
LD B,0 ; BC = word offset from ADSYM, graphics table
LD HL,ADSYM
ADD HL,BC
LD E,(HL)
INC HL
LD D,(HL) ; DE points to graphic
LD A,(CHLIN) ; A = row, 0..23
LD H,0
LD L,A ; HL = row, 0..23
ADD HL,HL ; x 32:
ADD HL,HL
ADD HL,HL
ADD HL,HL
ADD HL,HL ; HL = 32 * row
LD A,(COLUM)
LD B,0
LD C,A ; BC = column
ADD HL,BC ; HL = 32 * row + column
LD BC,22528 ; Start of attributes file
ADD HL,BC ; HL points to row & col in attr file
LD A,(TERSY) ; Terrain symbol
CP 6 ; Grass => no graphic
JR Z,NG0
PUSH HL
CALL G0 ; Display graphic
POP HL
LD A,(TERSY)
CP 10 ; Lake => no flag on horizon nor ground bar
JR Z,NGRAF

NG0: LD DE,31
ADD HL,DE ; HL points to start of ground (grass) in attr file
LD DE,62350 ; DE points to grass in ADS
CALL G0 ; Display grass
LD A,(HFLAG)
CP 0 ; No horizon flag => no more on this horizon point
JR NZ,NGRAF
LD DE,34
AND A
SBC HL,DE ; HL points to horizon flag position
LD A,H
CP 91 ; If off screen don't show it
JR NC,NGRAF
LD (HL),3BH ; Paper white, ink magenta

NGRAF: LD A,(COLUM) ; Get column
CP 28 ; If > 28, all done
JR NC,ENDIS
ADD A,3 ; Else next column

LD (COLUM),A
LD A,(DBEAR)
ADD A,8 ; Next horizon point bearing
LD (DBEAR),A
JP DLOO1 ; Next horizon point (loop)

ENDIS: LD HL,22895 ; Line 11, column 15 = centre of screen
SET 7,(HL) ; Flash = level marker
LD A,(TG) ; Get observer's terrain type (x 2)
LD HL,ADOBS ; Offset pointer from ADOBS, message table
LD C,A
LD D,0
ADD HL,DE
LD E,(HL)
INC HL
LD D,(HL) ; DE points to message string
LD HL,23264 ; Display at line 23 column 0
CALL T0 ; Display observer's terrain message
LD A,(OBSFE)
CP 0 ; If no feature at observer, then done
JR Z,ENDAL
CALL SHOFE ; Show feature
LD HL,(OBSCO)
LD A,32
CP H ; If no flag at observer, then done
JR Z,ENDAL
LD DE,64000 ; Point to graphic for flag
LD HL,23279 ; Line 23 column 15
CALL G0 ; Display flag

ENDAL: RET
HFLAG: DEFB 0 ; Whether flag is visible from horizon point

END

ORG 62040 ; HIGH - returns int (height) for test purposes
HEIGH: EQU 62737
LH: EQU 63011
HIGH: CALL HEIGH

LD BC,(LH)
RET
END

Part 6 - Turning and moving forward
Two modules:

SCROLLS - move screen left and right in character sized steps (not pixel sized
because colours would go funny; colours are in the attributes file which is character
based).

MOVE - moves the observer forward one pace (about 1m +/- 10%) and drifts the
bearing by up to +/- 1 b-degree.

In scrolling we have to move both the display file and the attributes file. The display
files is moved first, to minimise the period when the colours might be wrong.

ORG 61700 ; SCROLLS - left & right scrolling (panning)
TURN: EQU 62052
ATTR0: EQU 63850
BEAR: EQU 63431
RIGHT: LD A,(BEAR) ; Turn right 8 b-degrees

ADD A,8
LD (BEAR),A
LD B,3 ; Scroll left 3 times = 1 symbol space

LEFLO: PUSH BC
CALL SCLEF
POP BC
DJNZ LEFLO
JP TURN

LEFT: LD A,(BEAR) ; Turn left 8 b-degrees
SUB 8
LD (BEAR),A
LD B,3 ; Scroll right 3 times

RIGLO: PUSH BC
CALL SCRIG
POP BC
DJNZ RIGLO
JP TURN

BACK: LD A,(BEAR) ; Half turn, 128 b-degrees
ADD A,128
LD (BEAR),A

LD B,30 ; Scroll left 30 times
JR LEFLO

; Start addresses of display & attr files:
SDF: EQU 16384
SAF: EQU 22528
; Scroll left by 1 character:
SCLEF: LD HL,SDF

CALL CLCOL ; Clear left column
LD BC,6143 ; Bytes to move
LD DE,SDF ; Destination
LD HL,SDF+1 ; Source
LDIR ; Move display file
LD BC,767
LD DE,SAF
LD HL,SAF+1
LDIR ; Move attr file
RET

; Scroll right by 1 character:
SCRIG: LD HL,SDF+31

CALL CLCOL ; Clear right column
LD BC,6143
LD DE,SDF+6143
LD HL,SDF+6142
LDDR ; Move display file
LD BC,767
LD DE,SAF+767
LD HL,SDF+766
LDDR ; Move attr file
RET

; Clear column whose top is pointed to by HL:
CLCOL: LD B,192

LD DE,32
CL1: LD (HL),0 ; Zero display file (paper everywhere!)

ADD HL,DE
DJNZ CL1
LD A,(ATTR0) ; (Might depend on whether colour TV?)
LD B,24

CL2: LD (HL),A ; Reset attributes in attr file
ADD HL,DE
DJNZ CL2
RET
END

ORG 61850 ; MOVE - move observer forward and drift
BEAR: EQU 63431
TURND: EQU 62404
MUL212: EQU 62417
TERRA: EQU 62580
SINB: EQU 63432
COSB: EQU 63433
FXT: EQU 63434
LXT: EQU 63435
HXT: EQU 63436
FYT: EQU 63437
LYT: EQU 63438
HYT: EQU 63439
FXH: EQU 63440
LXH: EQU 63441
HXH: EQU 63442
FYH: EQU 63443
LYH: EQU 63444
HYH: EQU 63445
FHH: EQU 63446
LHH: EQU 63447
HHH: EQU 63448
FHG: EQU 63449
LHG: EQU 63450
HHG: EQU 63451
XF: EQU 63452
YF: EQU 63454
FINFL: EQU 63456
TG: EQU 63458
TERSY: EQU 63459
FEATY: EQU 62905
OBSFE: EQU 63463
CODE: EQU 62906
OBSCO: EQU 63464
DISPL: EQU 62055
MOVE: LD A,R ; Get random byte from Z80 refresh register

LD C,A ; treat as fraction to be multiplied by 0.1
LD D,0
PUSH DE ; Save fraction while bearing drifts
LD A,(BEAR)
BIT 5,C
JR Z,NPLUS
ADD A,1 ; Inc bearing by 1 b-degree

NPLUS: BIT 3,C
JR Z,NOMIN
SUB 1 ; Dec bearing by 1 b-degree

NOMIN: LD (BEAR),A
CALL TURND ; Revise sin & cos
POP DE ; Retrieve random fraction
LD C,1AH ; C = 0.1
CALL MUL212 ; HL = random part of distance to move
LD DE,0E6H ; DE = fixed part of same = 0.9
ADD HL,DE ; HL = distance to move

PUSH HL ; Save for incrementing yt
EX DE,HL ; Move to DE for incrementing xt
LD A,(SINB) ; sin(BEAR)
LD C,A
CALL MUL212 ; HL = sinb * distance to move
LD DE,(FXT) ; Adding to xt : get fractional & low parts of xt
LD A,(HXT) ; Get high part of xt
ADD HL,DE
ADC A,B ; Use sign byte from LU"!" result above
LD (FXT),HL ; xt = xt + move * sinb
LD (FXH),HL ; New xt will be used to find obeserver's height
LD (HXT),A
LD (HXH),A

; Similarly for yt:
POP DE
LD A,(COSB)
LD C,A
CALL MUL212
LD DE,(FYT)
LD A,(HYT)
ADD HL,DE
ADC A,B
LD (FYT),HL
LD (FYH),HL
LD (HYT),A
LD (HYH),A
LD HL,YF+1 ; Test for finish:
CP (HL) ; Do MSBs of yt & yf match?
JR NZ,HMOVE
LD A,(LYT)
AND 0FCH
LD B,A
DEC HL
LD A,(HL)
AND 0FCH
CP B ; And LSBs, within 4 metres?
JR NZ,HMOVE
DEC HL
LD A,(HXT)
CP (HL) ; And MSBs of xt and xf?
JR NZ,HMOVE
LD A,(LXT)
AND 0FCH
LD B,A
DEC HL
LD A,(HL)
AND 0FCH
CP B
JR NZ,HMOVE ; And LSBs within 4 metres?
LD A,1 ; If all matched we are at finish
LD (FINFL),A ; Set flag for BASIC shell to test

HMOVE: CALL TERRA ; Get observer's height and terrain
LD A,(FHH) ; Save height in HG

LD (FHG),A
LD HL,(LHH)
LD (LHG),HL
LD A,(TERSY) ; Save terrain type in TG
LD (TG),A
LD A,(FEATY) ; Save feature type
LD (OBSFE),A
LD HL,(CODE) ; Save flag code
LD (OBSCO),HL
JP DISPL

x0, y0 is top left, to draw the map in normal reading order, not bottom left as set up by
the user in the BASIC shell.

A cross-section is actually done by using the VIEW3D kernel with ny = 1 (ie, just 1
line to draw).

Part 7 - Maps
The first routine in this section, DRAWFN, is a subroutine for plotting or unplotting
points in a vertical line according to a binary pattern held in array BAR. This is used
to draw contoured cross-sections and 3D views (which are essentially repeated
cross-sections).

Maps are drawn by calling a standard harness (entry FHEAD) having first poked the
address of a kernel routine into MADDR from the BASIC shell. Different types of map
are produced by different kernels. The kernel determines the map content for a
particular point (x, y), while the harness scans a rectangular area, the definition of
which has been poked from the BASIC shell:

ORG 61500 ; DRAWFN subroutine
PLOT: EQU 22E5H ; Spectrum Rom routine. Enter with B = y, C = x
BAR: DEFS 100 ; Standard contoured line, preset
X0: DEFB 0 ; Screen position for
Y0: DEFB 0 ; start of line
LEN: DEFB 0 ; Length to plot
PBAR: Ld HL,BAR

LD A,(LEN)
LD D,A ; Counter
LD BC,(X0) ; BC = y0 * 256 + X0

PLOOP: LD A,(HL)
CP 0
JR Z,BLACK
SET 2,(IY+87) ; Inverse video (Spectrum system variable)
JR DO

BLACK: RES 2,(IY+87) ; Normal video (Spectrum system variable)
DO: PUSH HL

PUSH DE
PUSH BC
CALL PLOT
POP BC
POP DE
POP HL
INC B ; Y = y + 1. No change in x
LD A,175 ; If off top of screen
CP B
JR C,PEND ; then no more to plot
DEC D ; Count point just plotted
XOR A
CP D ; If LEN points done
JR Z,PEND ; then no more
INC HL ; Next point on bar
JR PLOOP

PEND: RET
END

ORG 61300 ; MAPS - harness, scans rectangular area to be mapped
X0: DEFW 0 ; Start of plot, top left on map
Y0: DEFW 0
XL: DEFW 0 ; Start of current line on map
YL: DEFW 0
XH: EQU 63441 ; Current position on map (global coordinates)
YH: EQU 63444
DXL: DEFW 0 ; Map increments along scan line, as diagram above
DYL: DEFW 0
DXC: DEFW 0 ; MAP increments between scans (ie, column)
DYC: DEFW 0
NX: DEFB 0 ; No of points in scan line, 0..255 or 0..31
NY: DEFB 0 ; No of lines, 0..191 or 0..23
XS: DEFB 0 ; Current screen x, 0..255
YS: DEFB 0 ; Current screen y, 191..0
XX: DEFB 0 ; Current screen character x, 0..31

YX: DEFB 0 ; Current screen character y, 0..23
MADDR: DEFW 0 ; Address of 1-point kernel for current map type -

; poke from BASIC first
FHEAD: LD HL,(X0) ; Start first line

LD (XL),HL
LD HL,(Y0)
LD (YL),HL
XOR A
LD (XH-1),A ; No fractional positions
LD (YH-1),A
LD A,(NY) ; No of lines
LD B,A ; B = line counter
LD A,191 ; Line start on screen (175 for FINEC)
LD (YS),A
XOR A
LD (YX),A

FLINE: PUSH BC ; Start any line
LD HL,(XL) ; Work point at line start
LD (XH),HL
LD HL,(YL)
LD (YH),HL
LD A,(NX) ; No of points in line
LD B,A ; B = point counter
XOR A
LD (XS),A ; Line start on screen
LD (XX),A

FPT: PUSH BC ; Do one point
LD HL,(MADDR) ; Get address of kernel for current map type
JP (HL) ; and go there

FRET: POP BC ; returning always to here
DJNZ NEXTP
POP BC
DJNZ NEXTL
RET

; Next point in a line:
NEXTP: LD HL,(XH)

LD DE,(DXL)
ADD HL,DE
LD (XH),HL
LD HL,(YH)
LD DE,(DYL)
ADD HL,DE
LD (YH),HL
LD HL,XX ; Increment character no
INC (HL) ; (to handle all map types)
LD HL,XS ; Increment pixel no
INC (HL)
JR FPT ; Do next point

; Next line:
NEXTL: LD HL,(XL)

LD DE,(DXC)
ADD HL,DE
LD (XL),HL

LD HL,(YL)
LD DE,(DYC)
ADD HL,DE
LD (YL),HL
LD HL,YX
INC (HL) ; Character lines count up from 0 at top
LD HL,YS
DEC (HL) ; Pixel lines count down from 191 at top
JR FLINE ; Do next line
END

Now for some kernels…

ORG 61240 ; VIEW3D - a map kernel - (MADDR) = VIEW
HEIGH: EQU 62737
LH: EQU 63011
X0: EQU 61600
Y0: EQU 61601
LEN: EQU 61602
PBAR: EQU 61603
XS: EQU 61318
YS: EQU 61319
FRET: EQU 61385
VIEW: CALL HEIGH ; (LH) = int (height (XH, YH))

LD HL,(LH)
LD BC,204 ; Lake height
AND A ; Clear carry
SBC HL,BC ; HL = HL - lake height
SRA H
RR L
SRA H
RR L
SRA H
RR L ; HL = HL / 8
LD A,L
CP 128 ; If < 0
JR C,HPOS
XOR A ; then A = 0

HPOS: INC A
LD (LEN),A
LD A,(YS) ; 191..0
SUB 70
LD (Y0),A ; (i0 in BASIC version)
LD A,(XS)
LD (X0),A
CALL PBAR
JP FRET ; End of kernel
END

The next one, FINEC, is very similar:

ORG 61240 ; FINEC - fine contours kernel - (MADDR) = FINEC
HEIGH: EQU 62737
PLOT: EQU 22E5H ; (In ROM)
BAR: EQU 61500
LH: EQU 63011
X0: EQU 61600
Y0: EQU 61601
LEN: EQU 61602
PBAR: EQU 61603
XS: EQU 61318
YS: EQU 61319
FRET: EQU 61385
VIEW: CALL HEIGH ; (LH) = int (height (XH, YH))

LD HL,(LH)
LD BC,204 ; Lake height
AND A ; Clear carry
SBC HL,BC ; HL = HL - lake height
SRA H
RR L
SRA H
RR L
SRA H
RR L ; HL = HL / 8
LD A,L
CP 128 ; If < 0
JR C,HPOS
XOR A ; then A = 0

HPOS: INC A
LD H,0
LD L,A ; 191..0
LD BC,BAR
ADD HL,BC ; HL indexes BAR array
LD A,(HL) ; Value to plot, from BAR
CP 0
JP Z,FRET ; IF white, no plot
LD BC,(XS) ; (XS) in C, (YS) in B
CALL PLOT
JP FRET ; End of kernel
END

To make fine contours map from BASIC:

8500 LET NX=255:LET NY=175:POKE 61322,72:POKE 61323,238:
POKE 61348,175:GOTO 9500

and

6000 CLS:INPUT “Fine or coarse map? [f/c]”;F$:IF F$=”f” THEN GOTO 8500
6100 LET NX=32:LET NY=24:POKE 61322,187:POKE 61323,238:GOTO 9500



The Forest - ZX Spectrum page 53

The next chunk of code includes 2 kernels. CMAP is the coarse contour map and
TMAP is the terrain map.

ORG 61088 ; CMAP & TMAP kernels
HEIGH: EQU 62737
LH: EQU 63011
FRET: EQU 61385
TERRA: EQU 62580
TATTR: EQU 63745
T0: EQU 63757
SAF: EQU 22528 ; Start of attr file
XX: EQU 61320 ; Character column, 0..31
YX: EQU 61321 ; Character row, 0..23
LXH: EQU 63442 ; Global position on map
LYH: EQU 63445
TERSY: EQU 63459
SYMS: DEFM "r"

DEFB 56 ; Black on white
DEFM "t"
DEFB 32 ; Black on green

DEFM "m"
DEFB 31 ; White on magenta
DEFM "g"
DEFB 48 ; Black on yellow
DEFM "s"
DEFB 7 ; White on black
DEFM "l"
DEFB 15 ; White on blue

NRIGH: DEFB 5 ; No of right shifts to scale height (must not be 0)
CHAR: DEFB 0

DEFB 0 ; End of CHAR string, for T0
CMAP: CALL HEIGH

LD DE,(LH)
LD A,D
CP 0
JR NZ,SHIFT
LD A,E
CP 204 ; Lake height
JR NC,SHIFT
LD E,200 ; If lake replace with surface (as in TERRA)

SHIFT: LD A,(NRIGH)
LD B,A

CLOOP: SRL D
RR E ; E = height to plot
DJNZ CLOOP
LD A,'<' ; Lowest height band "letter"
ADD A,E
AND 7FH ; No characters above 128!
LD (CHAR),A ; To display using T0
AND 7 ; Paper colour
ADD A,A
ADD A,A
ADD A,A ; Moves paper colour to bits 3, 4, 5
CP 40 ; 101000 - check contrast for letter
JR NC,SHOW ; Ink black
OR 7 ; Ink white

SHOW: LD (TATTR),A
LD A,(YX)
LD L,A
LD H,0
ADD HL,HL
ADD HL,HL
ADD HL,HL
ADD HL,HL
ADD HL,HL ; 5 times: HL = HL * 32
LD A,(XX)
LD C,A
LD B,0
ADD HL,BC
LD BC,SAF
ADD HL,BC ; HL points to attr file position for T0
LD DE,CHAR ; Point to string to display
CALL T0



The Forest - ZX Spectrum page 55

JP FRET ; End of CMAP kernel
TMAP: LD HL,(LXH) ; Terrain map kernel

LD (TEMPX),HL ; Save coordinates because FEATU alters them
LD HL,(LYH)
LD (TEMPY),HL
CALL TERRA
LD HL,(TEMPX)
LD (LXH),HL
LD HL,(TEMPY)
LD (LYH),HL
LD A,(TERSY)
LD C,A
LD B,0
LD HL,SYMS
ADD HL,BC
JP TEND

; Rest of TMAP code went to start of module because out of space to next
; module

ORG 61088
TEND: LD A,(HL) ; Character to display

LD (CHAR),A
INC HL
LD A,(HL) ; Attributes
JR SHOW

TEMPX: DEFW 0
TEMPY: DEFW 0

END

Part 9 - The BASIC shell
I have not retyped this. You might get the digital version from World of Spectrum (the
page for The Forest) because they have TAP format files for use with Spectrum
emulators. They should enable the BASIC source to be seen once loaded. This
would not work for assembly files of course because only the assembled machine
code is present. Even if you disassembled that you would not have much clue as to
how it works without my comments.

https://worldofspectrum.org/archive/software/games/the-forest-phipps-associates

Memory map for assembly parts of The Forest
See also page 24.

Further information
The Forest on the World of Spectrum web site.

A 1984 review of the original version, in Crash magazine.

https://worldofspectrum.org/archive/software/games/the-forest-phipps-associates
https://www.crashonline.org.uk/01/forest.htm

